# =======Create_And_Transient_Simulate_1F_Noise=======

## SIMULATE 1/F NOISE DURING TRANSIENT ANALYSIS.







```
==================Want_100_1us_steps================================
Total_Period_s =   0.1
Bin_Resolutio_Hz = 10
Sample_Period_s =  0.001
Nyquist_Hz =       500
==================Create_PWL_array_and_Index_and_Plot==============
==================Add_1Vrms_Noise_to_PWL_array=====================
==================Install_the_PWL_array============================
==================Run_and_Plot=====================================
==================Find_Ave_Rms=====================================
Average level      -0.022418
RMS level          0.783903
==================FFT_and_Plot=====================================
FFT_BandWidth_Hz=  1000
FFT_resolution_Hz= 10
==================done==============================================
```

### Create_1F_Noise

```
*================Need_Voltage_Sources_to_alter====================="
V1                V1      0       0     dc
V2                V2      0       0     dc
V3                V3      0       0     dc
V4                V4      0       0     dc
V5                V5      0       0     dc
BVOUT             VOUT    0       V =   V(V1)+ V(V2)+ V(V3) + V(V4) + V(V5)
.control
set               pensize = 2
echo              "==================Want_100_1us_steps================================"
let n =           100
let tstep =       1ms
let period_t =    n*tstep
let Bin_Hz =      1/period_t
let nyquist =     .5/tstep
echo              "Total_Period_s =   $&period_t"
echo              "Bin_Resolutio_Hz = $&Bin_Hz"
echo              "Sample_Period_s =  $&tstep"
echo              "Nyquist_Hz =       $&nyquist"
echo              "==================Create_PWL_array_and_Index_and_Plot=============="
let pwl_1 =       vector(2*n)*tstep*0.5
let n2  =         50
let pwl_2 =       vector(2*n2)*tstep
let n3  =         25
let pwl_3 =       vector(2*n2)*tstep*2
let n4  =         12
let pwl_4 =       vector(2*n2)*100m/24
let n5  =         6
let pwl_5 =       vector(2*n2)*100m/12
let ii =          vector(2*$&n)
echo              "==================Add_1Vrms_Noise_to_PWL_array====================="
let index =       0
repeat            $&n
let               pwl_1[1+2*index] = 1*(rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)-507.5)/102.879
let index =       index + 1
end
let index =       0
repeat            $&n2
let               pwl_2[1+2*index] =  1.4*(rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)-507.5)/102.879
let index =       index + 1
end
let index =       0
repeat            $&n3
let               pwl_3[1+2*index] =  2*(rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)-507.5)/102.879
let index =       index + 1
end
let index =       0
repeat            $&n4
let               pwl_4[1+2*index] =  2.8*(rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)-507.5)/102.879
let index =       index + 1
end
let index =       0
repeat            $&n5
let               pwl_5[1+2*index] =  4*(rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)-507.5)/102.879
let index =       index + 1
end
```

```
echo              "==================Install_the_PWL_arrays============================="
alter             @v1[pwl] = pwl_1
alter             @v2[pwl] = pwl_2
alter             @v3[pwl] = pwl_3
alter             @v4[pwl] = pwl_4
alter             @v5[pwl] = pwl_5
echo              "==================Run_and_Plot======================================="
let period_s =    tstep/2
*tran             0.5us 100us
tran              $&period_s $&period_t   0  $&period_s
plot              v1 v2 v3 v4 v5 vout
*plot             v(V1) pointplot
echo              "==================Find_Ave_Rms======================================="
let averVal =     mean(v1)
let noisAC =      v1 - averVal
let RmsVal =      sqrt(mean(noisAC* noisAC))
echo              "Average level      $&averVal"
echo              "RMS level          $&RmsVal"
unlet averVal
unlet RmsVal
echo              "==================FFT_and_Plot======================================="
linearize
let               FFT_BandWidth_Hz =  1k
let               FFT_resolution_Hz = 10
echo              "FFT_BandWidth_Hz=  $&FFT_BandWidth_Hz"
echo              "FFT_resolution_Hz= $&FFT_resolution_Hz"
set               specwindow =   "rectangular"
spec              $&FFT_resolution_Hz  $&FFT_BandWidth_Hz   $&FFT_resolution_Hz     v(vout)
let nyquist_V =   1/(1 +(frequency/600)*(frequency/600)*(frequency/600)*(frequency/600)*(frequency/600))
let expect_V =    3*nyquist_V/sqrt(frequency)
plot              mag(vout) nyquist_V  expect_V  loglog
echo              "==================done==============================================="
.endc
.end
```